TP : Courbes paramétrées

1 Tracé d'une courbe explicite $y = f(x)$

Le tracé d'une courbe dans SCILAB repose sur un principe très simple :

1. on crée un vecteur des abscisses, par exemple par l'instruction :

   `x = linspace(-3,3,10);`

   qui crée 10 points également espacés dans l'intervalle $[-3, 3]$. Pour s'en rendre compte on pourra dans un premier temps ne pas mettre le ";". On a donc x qui est le vecteur :

   $$x_1 = -3, x_2 = -2.3333333, \cdots, x_{10} = 3$$

2. Ensuite nous construisons par l'instruction :

   `y = sin(x);`

   le vecteur y des ordonnées :

   $$y_1 = \sin(x_1), y_2 = \sin(x_2), \cdots, y_{10} = \sin(x_{10})$$

   obtenue par application de la fonction sinus à chacun des éléments x_i du vecteur "x".

3. Enfin, l'instruction : `plot(x,y,'b-')` va créer une ligne brisée joignant les points (x_i, y_i) entre eux.

Résumons notre programme SCILAB que nous allons sauvegarder dans le fichier "trace1.sce" :

```
x = linspace(-3,3,10);
y = sin(x);
plot(x,y,'b-')
```

On trouvera le résultat du tracé dans la fenêtre graphique de SCILAB. Evidemment, le résultat n'est pas très bon. Il faut donc augmenter le nombre de points de subdivision :

```
n = 1000;
x = linspace(-3,3,n);
y = sin(x);
plot(x,y,'b-')
```

Noter au passage que le graphique se superpose au graphique précédent.

La commande `clf` efface la fenêtre graphique courante.
La commande `xdel(winsid())` supprime toutes les fenêtres graphiques.
L'instruction `figure()` crée une nouvelle fenêtre graphique.
L'instruction `set(gca(),"isoview","on")` normalise l'échelle en abscisse et en ordonnée.

Exercice 1 *Tester l'exemple précédent.*

Exercice 2 *Tracer la courbe représentative de la fonction :*

$$x \mapsto x^2, \quad x \in [-2, 2].$$

Exercice 3 *Tracer la courbe représentative de la fonction :*

$$x \mapsto e^{-x} \sin(10\,x), \quad x \in [0, 2].$$

Remarque : On notera que exp(-x) crée un vecteur de la même taille que x, ainsi que sin(10*x). Pour la multiplication terme à terme de ces deux vecteurs on utilise le produit `.*` au lieu de `*`.

2 Courbes implicites

• — **L'ellipse** de demi-axes $a > 0$ et $b > 0$ est définie par l'équation implicite

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1\,.$$



Exercice 4 *Tracer cette courbe pour $a = 5$ et $b = 2$, en se ramenant au cas précédent, c'est-à-dire en considèrant cette ellipse comme la réunion des 2 courbes explicites $y = f_1(x) = \frac{b}{a}\sqrt{a^2 - x^2}$ et $y = f_2(x) = -\frac{b}{a}\sqrt{a^2 - x^2}$ pour $x \in [-a, a]$ (en différenciant les tracés).*



• — **Le Folium de Durer**, défini par l'équation implicite

$$(x^2 + y^2)\left(2\,(x^2 + y^2) - a^2\right)^2 - a^4\,x^2 = 0\,,$$

ne pourra pas être traité à l'identique (ici : $a = 10$).

Cependant, on peut remarquer que l'intersection de cette courbe avec la droite *variable*, passant par l'origine, $y = t\,x$ (le paramètre t définissant la pente de la droite) conduit à l'équation bi-carrée :

$$4\,(1 + t^2)^3\,X^2 \;-\; 4\,a^2\,(1 + t^2)^2\,X \;+\; t^2\,a^4 = 0$$

avec $X = x^2$ et bien sur $y = t\,x$ (en omettant l'origine).

Exercice 5 *Compléter le programme suivant permettant de déterminer les intersections de cette courbe avec les droites $y = t_i\,x$, $t_i = \tan(\theta_i)$ pour des valeurs de θ_i comprises entre $-\pi/2$ et $\pi/2$.*

```
// RESOLUTION de l'equation A X^2 + B X + C
function [x1,x2] = Resol2(A,B,C)
    delta = B^2 - 4*A*C;
    if (delta>= 0) then
        x1 = (-B-sqrt(delta))/(2*A);
        x2 = (-B+sqrt(delta))/(2*A);
    else
        error('pas de racine')
    end
endfunction
```



```
// FOLIUM DE DURER : tracé par intersection
// avec la droite variable y = tx
a = 10;
pas = %pi/20;
for theta = -%pi/2+0.01 : pas : %pi/2-0.01
    t = tan(theta);
    A = ....;
    B = ....;
    C = ....;
    [x1,x2] = Resol2(A,B,C);
    if x1 >= 0 then
        x = sqrt(x1); y = t*x;
        plot(x,y,'bo')
        plot(-x,-y,'bo')
    end
    if x2 >= 0 then
        x = ....; y = ....;
        plot(.....)
        plot(.....)
    end
end
```

Exercice 6 [Bonus] *Appliquer la même idée pour la* **lemniscate de Bernouilli** *définie par l'équation implicite*

$$(x^2 + y^2)^2 = a^2\,(x^2 - y^2)$$

Nous verrons ci-dessous que ces courbes admettent une représentation paramétrique plus commode pour le tracé.

3 Les courbes paramétrées

La courbe paramétrée,

$$\left\{\begin{array}{l} x = f(t) \\ y = g(t) \end{array}\right. , \quad t \in [a,b]$$

est constituée de l'ensemble des points du plan $\{(f(t), g(t)),\ t \in [a,b]\}$. Le graphe d'une fonction correspond ainsi au cas $f(t) = t$.

• — Le programme "type"

Un programme SCILAB de base pour tracer la courbe paramétrée $\{(f(t), g(t)),\ t \in [a,b]\}$ peut s'écrire schématiquement sous la forme :

```
// Tracé courbe paramétrée
function x = f(t)
    x = ...;
endfunction

function y = g(t)
    y= ...;
endfunction

a = ....; b = ....;
n = ....;
T = linspace(a,b,n);
X = f(T);
Y = g(T);
plot(X,Y,...);
```

les éléments signalés par "..." étant à compléter.

Exercice 7 *Tracer le cercle d'équations paramétriques* $x = 3\ \cos(t), y = 3\ \sin(t)$*,* $t \in\ [0, 2\pi]$*, puis l'ellipse d'équations paramétriques* $x = 5\ \cos(t), y = 2\ \sin(t)$*,* $t \in\ [0, 2\pi]$*.*

Exercice 8 *Tracer la courbe paramétrée* $\mathcal{C} = \{(x(t), y(t)),\ t \in [a,b]\}$ *avec*

$$\left\{\begin{array}{l} x(t) = 2\sin(t)^2\cos(t) \\ y(t) = 2\cos(t)^2\sin(t) \\ t \in [0, 2\pi] \end{array}\right.$$

• — Quelques courbes paramétrées remarquables

Exercice 9 Courbes de Lissajous produites par les oscilloscopes :

$$\begin{array}{rl} x(t) = & \cos(k\,t) \\ y(t) = & \sin(h\,t) \end{array}$$

Tracer cette courbe pour différentes valeurs entières de k *et* h*. On pourra tester les couples* $(k, h) = (1,2), (3,2), (3,4), (5,6), (9,8), \cdots$ *et on choisira* $t \in [0, p\,\pi]$*, avec* p *entier à tester.*

Exercice 10 *Tracer la courbe paramétrée* $\mathcal{C} = \{(x(t), y(t)),\ t \in [a,b]\}$ *avec*

$$\left\{\begin{array}{l} x(t) = sin(5t) \\ y(t) = sin(6t) \\ t \in [0, 2\pi] \end{array}\right.$$

Exercice 11 ***Les cycloïdes :***

$$\begin{array}{l} x(t) = t - r\ \sin(t) \\ y(t) = 1 - r\ \cos(t) \end{array}$$

Tracer cette courbe en faisant varier la valeur de r *:* $r = 1$ *(la cycloïde classique), puis* $r < 1$*, et* $r > 1$*. On choisira par exemple* $t \in [0, 2\pi]$*, puis* $t \in [0, 4\pi]$*,...*

Exercice 12 ***Le Folium de Durer :*** *(que l'on retrouve donc ici sous forme paramétrique)*

$$\begin{aligned} x &= \tfrac{a}{2}\big(\cos(t) + \cos(3\,t)\big) \\ y &= \tfrac{a}{2}\big(\sin(t) + \sin(3\,t)\big) \end{aligned}$$

Tracer cette courbe pour $a = 10$ et $t \in [0, 2\pi]$.

• — Influence du paramétrage

⋄ On considère les deux paramétrisations suivantes du cercle unité (privé d'un point).

$$\left\{ \begin{array}{l} x_1(t) = \cos(t) \\[2ex] y_1(t) = \sin(t) \end{array} \right. \quad t \in]-\pi, \pi[\qquad \text{et} \qquad \left\{ \begin{array}{l} x_2(t) = \dfrac{1-t^2}{1+t^2} \\[2ex] y_2(t) = \dfrac{2\,t}{1+t^2} \end{array} \right. \quad t \in]-\infty, +\infty[$$

Exercice 13 *Tracer la première courbe paramétrée pour $t \in I_1 = [-5\pi/6, 5\pi/6]$. Tracer ensuite la deuxième courbe paramétrée pour $t \in I_2 = [-a, a]$, avec $a = \sqrt{(2+\sqrt{3})/(2-\sqrt{3})}$.*
Tracer enfin l'image de 30 points équirépartis dans chacun des intervalles I_1 et I_2 — voir figure ci-dessous. Commentaires ?



⋄ On considère maintenant la **Lemniscate de Bernouilli** définie par la représentation paramétrique suivante

$$M(t) = \left\{ \begin{array}{l} x = a\,\dfrac{t+t^3}{1+t^4} \\[3ex] y = a\,\dfrac{t-t^3}{1+t^4} \end{array} \right. \qquad t \in]-\infty, +\infty[$$



Exercice 14 *Compléter le programme ci-dessous permettant de tracer cette lemniscate pour $t \in [-5, 5]$, dans un premier temps avec* **200** *valeurs t_i, puis ensuite avec seulement* **30** *valeurs t_j. On tracera par ailleurs en chacun de ces derniers points $M(t_j)$ le vecteur dérivé $M'(t_j)$ à l'aide de la fonction* PlotVecteur(A,u) *ci-jointe (il sera judicieux d'utiliser un coefficient d'atténuation pour la norme de ces vecteurs). Analyser le lien entre la norme des vecteurs dérivés (cinématique du paramétrage) et la précision du tracé.*

```
//LEMNISCATE DE BERNOUILLI
clf
a = 5;
// affichage avec 200 valeurs dans [-a,a]
n = 200;
t = linspace(-a,a,n);
x = a * (t + t.^3)./(1 + t.^4);
y = ....;
plot(x,y, '-b')
// affichage avec 30 valeurs dans [-a,a]
n = 30;
u = linspace(-a,a,n);
x = a * (u + u.^3)./(1 + u.^4);
y = ....;
plot(x,y, '-ro')
set(gca(),"isoview","on")
// Vecteur dérivé :
dx = a * ((1 + 3*u.^2).*(1 + u.^4) - 4*(u.^3).*(u + u.^3)) ./ ((1 + u.^4).^2);
dy = ....;
// Tracé des vecteurs dérivés
// coef = coefficient d'atténuation
coef = 1/3;
for i = 1 : size(u,2)
    M = [x(i) y(i)];
    dM = coef * [dx(i) dy(i)];
    PlotVecteur(...,...)
    //PlotSerretFrenet(...,...)
end
```

```
// FONCTIONS
function PlotSerretFrenet(M,dM)
    // trace le repère de Serret-Frenet au point M
    // de la courbe
    // dM est le vecteur dérivé de la courbe en M
    // on commence par normaliser le vecteur dM
    dM = (1/sqrt(dM(1,1)^2 + dM(1,2)^2))*dM;
    T = M + dM;
    dN = [-dM(1,2) dM(1,1)];
    N = M + dN;
    plot(M(1,1),M(1,2),'bo')
    plot([M(1,1) T(1,1)],[M(1,2) T(1,2)],'-b')
    plot([M(1,1) N(1,1)],[M(1,2) N(1,2)],'-b')
endfunction

function PlotVecteur(A,u)
    // trace le vecteur u a partir du point A
    B = A + u;
    plot([A(1,1) B(1,1)],[A(1,2) B(1,2)],'-r')
    plot(A(1,1),A(1,2),'ro')
endfunction
```

• — Repère mobile de Serret Frenet

Exercice 15 *On considère à nouveau la lemniscate de Bernouilli pour $t \in [-5,5]$. Compléter le programme précédent afin de tracer le repère de Serret Frenet en chacun des points $M(t_j)$ (échantillonnage de 30 points). On utilisera la fonction* PlotSerretFrenet(M,dM) *donnée ci-dessus.*



4 Courbes en coordonnées polaires

La description d'une courbe en *coordonnées polaires* consiste à donner la distance ρ de chacun de ses points M à l'origine O en fonction de l'angle θ que fait la demi-droite $[0M)$ avec l'axe des abscisses, autrement dit à donner la distance $\rho = \rho(\theta)$, $\theta \in [a,b]$. Ainsi, l'équation d'un cercle centré à l'origine en coordonnées polaires est : $\rho = \rho(\theta) = r,\ \theta \in [0,\ 2\pi]$.

Une courbe en polaire $\rho = \rho(\theta),\ \theta \in [a,b]$, peut se mettre sous forme paramétrique :

$$\left\{ \begin{array}{l} x(\theta) = \rho(\theta)\, \cos(\theta) \\ y(\theta) = \rho(\theta)\, \sin(\theta) \end{array} \right. \qquad \theta \in [a,b].$$

Exercice 16 *Tracer les courbes suivantes définies en coordonnées polaires (attention aux problèmes d'échelle !) :*

1. *La spirale logarithmique : $\rho(\theta) = e^{\theta}$,*
2. *la spirale d'Archimède : $\rho(\theta) = \alpha\ \theta$, où α est une constante à choisir,*
3. *la courbe*

$$\left\{ \begin{array}{l} \rho(\theta) = sin(2\theta)^2 + (sin(4\theta)^4)/2 \\ \theta \in [0, 2\pi] \end{array} \right.$$

4. la courbe

$$\begin{cases} \rho(\theta) = 1 + \varepsilon\cos(n\theta) \\ \theta \in [0, 2\pi] \end{cases} \quad \textit{dans les cas suivants :} \quad \begin{cases} \varepsilon = 0.5,\ n = 5,\ \theta \in [0, 2\pi] \\ \varepsilon = 0.5,\ n = 4.5,\ \theta \in [0, 4\pi] \\ \varepsilon = 1.5,\ n = 5,\ \theta \in [0, 2\pi] \\ \varepsilon = 1,\ n = \pi,\ \theta \in [0, 100\pi] \\ \varepsilon = 1,\ n = \pi,\ \theta \in [0, 1000\pi] \end{cases}$$

Exercice 17 *Tracer la courbe* $\rho(\theta) = 2/\cos(\theta)$, $\theta \in]-\pi/2, \pi/2[$. **Bonus** *: donner la représentation en coordonnées polaires de la droite d'équation* $y = x - 1$, *et tracer cette droite.*

5 Quelques mots sur le tracé d'une surface

Nous avons vu que le tracé d'une courbe représentative de $y = f(x),\ x \in [a, b]$ en dimension 2 consistait à découper l'intervalle $[a, b]$ avec n points équirépartis, à évaluer la fonction f en ces points, puis à fournir le résultat de ces calculs à l'instruction plot.

Le tracé d'une surface explicite $z = f(x, y),\ x \in [a, b],\ y \in [\alpha, \beta]$ relève du même principe. On découpe chacun des intervalles $[a, b]$ et $[\alpha, \beta]$ avec n points équirépartis. On évalue la fonction z en chacun des points $(x(i), y(j))$ dans un tableau de dimension $n \times n$ puis on appelle la fonction plot3d(x,y,z).

On donne ci-dessous à gauche l'exemple du tracé du paraboloïde :

$$z = x^2 + y^2,\ x \in [-1, 1],\ y \in [-1, 1].$$

```
// Paraboloide
clf
n = 50;
// Découpage des intervalles en x et y
x = linspace(-1,1,n);
y = linspace(-1,1,n);
// Formatage du tableau des z
z = zeros(n,n);
// Evaluation de la fonction z
for i = 1 : n
    for j = 1 : n
        z(i,j) = x(i)^2 + y(j)^2;
    end
end
// Tracé :
plot3d(x,y,z)
```

Exercice 18 *Tracer la surface :*

$$z = \exp(-x^2 - y^2),\ x \in [-2, 2],\ y \in [-2, 2].$$

Exercice 19 *Tracer la surface :*

$$z = \exp\left(-(x+1)^2 - (y+1)^2\right) + \exp\left(-(x-1)^2 - (y-1)^2\right),$$

$$x \in [-2, 2],\ y \in [-2, 2].$$

6 Un exemple de tracé d'une courbe 3D paramétrée

Exercice 20 *Exécuter le script scilab suivant, que l'on modifiera ensuite pour tracer d'autres courbes 3D.*

```
a = 0;
b = 2*%pi;
n = 400;
t = linspace(a,b,n);
x = sin(2*t);
y = sin(3*t);
z = cos(3*t);
param3d(x,y,z)
e = gce()
e.foreground = color('red');
```

Par exemple, on tracera l'hélice circulaire de rayon R *et de pas* a *:*

$$\begin{cases} x = R\cos(t) \\ y = R\sin(t) \\ z = (2\pi/a)\,t \end{cases} \qquad t \in [0, 2k\pi], \quad k \in \mathbb{Z}$$

On trouvera de nombreux exemples de courbes et surfaces à l'adresse : http ://www.mathcurve.com/